Q:

excel formula parse string by LAST "\" and ".pdf"

Is there a way to parse a string for the last child node of a path using a formula?
Ie. 
Stack\Over\Flow.pdf
parsing would get just "Flow".
=mid(a1,find(":",a1)+1,len(a1)) I've tried this but I have to specify the name "Flow" , but if the strings change for each path, it won't work.

A:

To find the last \ we use
FIND("}}}",SUBSTITUTE(A1,"\","}}}",LEN(A1)-LEN(SUBSTITUTE(A1,"\",""))))

Now we use MID to return everything after that:
MID(A1,FIND("}}}",SUBSTITUTE(A1,"\","}}}",LEN(A1)-LEN(SUBSTITUTE(A1,"\",""))))+1,99)

Then we substitute .pdf for a null string:
=SUBSTITUTE(MID(A1,FIND("}}}",SUBSTITUTE(A1,"\","}}}",LEN(A1)-LEN(SUBSTITUTE(A1,"\",""))))+1,99),".pdf","")

